

## 9/3.13

# Jeux de chevaux à deux joueurs

---

Vous connaissez très certainement ce jeu de société dans lequel des chevaux doivent parcourir un terrain en forme de plus pour se diriger vers la case centrale du jeu. En voici une version informatique écrite en Locomotive Basic.

Comme nous le verrons par la suite, le programme présenté est une sous-version que vous devrez améliorer.

## COMMENT FONCTIONNE LE PROGRAMME

Saisissez le listing ci-dessous :

```

1000 REM =====
1010 REM Jeu de chevaux a deux joueurs
1020 REM =====
1030 REM
1040 GOSUB 2000 'Affichage du jeu
1050 GOSUB 4000 'Jeu
1060 GOSUB 5000 'Commentaires sur la partie
1070 END
1080 '
2000 REM -----
2010 REM Affichage du jeu de depart
2020 REM -----
2030 REM
2040 MODE 1
2050 '
2060 ' - - - - -
2070 ' Caracteres graphiques utilises
2080 ' - - - - -
2090 '
2100 c1$=CHR$(150) 'Coin superieur gauche
2110 c2$=CHR$(154) 'Trait horizontal
2120 c3$=CHR$(156) 'Coin superieur droit
2130 c4$=CHR$(149) 'Trait vertical
2140 c5$=CHR$(147) 'Coin inf2rieur gauche
2150 c6$=CHR$(153) 'Coin inferieur droit
2160 c9$=CHR$(227) 'point sur le de
2170 DIM t(48),tp(12) 'Tableaux de jeu
2180 DIM x(48),y(48) 'Tableaux des positions
2190 '
2200 ' - - - - -
2210 ' Trace du circuit
2220 ' - - - - -
2230 '
2240 PEN 1
2250 s1$=c1$+STRING$(7,c2$)+c3$
2260 s2$=c4$+" . . . "+c4$
2270 s3$=c1$+STRING$(9,c2$)+c6$+" . . . "+c5$+STRING$(9,c2$)
+c3$
2280 s4$=c5$+STRING$(9,c2$)+c3$+" . . . "+c1$+STRING$(9,c2$)
+c6$
2290 s5$=c4$+" . . . "+c4$
2300 s6$=c5$+STRING$(7,c2$)+c6$
2310 s7$=c4$+" . . . . . . . . . "+c4$
2320 LOCATE 15,1
2330 PRINT s1$
2340 FOR i=1 TO 4
2350   LOCATE 15,i+1

```

```
2360 PRINT s2$
2370 NEXT i
2380 LOCATE 5,6
2390 PRINT s3$
2400 FOR i=1 TO 3
2410 LOCATE 5,i+6
2420 PRINT s7$
2430 NEXT i
2440 LOCATE 5,10
2450 PRINT s4$
2460 FOR i=1 TO 4
2470 LOCATE 15,i+10
2480 PRINT s5$
2490 NEXT i
2500 LOCATE 15,15
2510 PRINT s6$
2520 '
2530 ' - - - - -
2540 ' Trace des ecuries
2550 ' - - - - -
2560 '
2570 PEN 2
2580 s1$=c1$+STRING$(4,c2$)+c3$
2590 s2$=c4$+SPACE$(4)+c4$
2600 s3$=c5$+STRING$(4,c2$)+c6$
2610 LOCATE 7,11
2620 PRINT s1$
2630 FOR i=1 TO 3
2640 LOCATE 7,i+11
2650 PRINT s2$
2660 NEXT i
2670 LOCATE 7,15
2680 PRINT s3$
2690 LOCATE 8,12
2700 PRINT"o o"
2710 LOCATE 8,14
2720 PRINT"o o"
2730 PEN 3
2740 LOCATE 26,1
2750 PRINT s1$
2760 FOR i=1 TO 3
2770 LOCATE 26,i+1
2780 PRINT s2$
2790 NEXT i
2800 LOCATE 26,5
2810 PRINT s3$
2820 LOCATE 27,2
```

```

2830 PRINT"x  x"
2840 LOCATE 27,4
2850 PRINT"x  x"
2860 PEN 1 : LOCATE 19,8 : PRINT"A"
2870 PEN 2 : LOCATE 17,14 : PRINT"D"
2880 PEN 3 : LOCATE 21,2 : PRINT"D"
2890 '
2900 ' - - - - -
2910 ' Tableau de bord
2920 ' - - - - -
2930 '
2940 LOCATE 1,18
2950 PEN 1 : PRINT STRING$(40,c2$)
2960 LOCATE 1,20
2970 PRINT"Jet de des"
2980 LOCATE 17,20
2990 PEN 2 : PRINT"Joueur 1"
3000 LOCATE 31,20
3010 PEN 3 : PRINT"Joueur 2"
3020 PEN 1
3030 FOR i=1 TO 2
3040   FOR j=1 TO 4
3050     LOCATE 13+(i-1)*15,j+19
3060     PRINT c4$
3070   NEXT j
3080 NEXT i
3090 '
3100 ' - - - - -
3110 ' Memorisation des cases de deplacement
3120 ' - - - - -
3130 '
3140 FOR i=1 TO 48
3150   READ x(i),y(i)
3160 NEXT i
3170 DATA 17,14,17,13,17,12,17,11,17,10,17,9,15,9,13,9,11,9,
9,9,7,9
3180 DATA 7,8,7,7,9,7,11,7,13,7,15,7,17,7,17,6,17,5,17,4,17,
3,17,2
3190 DATA 19,2,21,2,21,3,21,4,21,5,21,6,21,7,23,7,25,7,27,7,
29,7,31,7
3200 DATA 31,8,31,9,29,9,27,9,25,9,23,9,21,9,21,10,21,11,21,
12,21,13
3210 DATA 21,14,19,14
3220 RETURN
4000 REM -----
4010 REM Deroulement d'une partie
4020 REM -----

```

```
4030 REM
4040 PEN 1
4050 jo=1
4060 c2=1 : d2=1 'Un cheval dans chaque camp
4070 FOR i=21 TO 25
4080   LOCATE 3,i:PRINT SPACE$(10);
4090 NEXT i
4100 LOCATE 5,17
4110 IF jo=1 THEN PEN 2:PRINT"Joueur 1, appuyez sur une touc
he"
4120 IF jo=2 THEN PEN 3:PRINT"Joueur 2, appuyez sur une touc
he"
4130 PEN 1
4140 a$=INKEY$:IF a$="" THEN 4140
4150 LOCATE 5,17
4160 PRINT SPACE$(32)
4170 GOSUB 6000 'Jet de de
4180 IF (de=6) AND (c2<>0) AND (jo=1) THEN so=1 'Sortie chev
al 1
4190 IF (de=6) AND (d2<>0) AND (jo=2) THEN so=5 'Sortie chev
al 5
4200 LOCATE 5,17
4210 PEN jo+1
4220 PRINT"Tapez J pour jouer et P pour passer"
4230 PEN 1
4240 a$=INKEY$:IF a$="" THEN 4240
4250 a$=UPPER$(a$)
4260 LOCATE 1,17
4270 PRINT SPACE$(40)
4280 IF a$="P" THEN IF jo=1 THEN jo=2 ELSE jo=1
4290 IF a$="J" THEN GOSUB 7000 'Deplacement
4300 IF fin=0 THEN 4100 'Boucle de jeu
4310 RETURN
5000 REM -----
5010 REM Commentaire sur la partie
5020 REM -----
5030 REM
5040 CLS
5050 PEN 1
5060 IF (14-f1=8) THEN PRINT" Le joueur 1 gagne la partie" E
LSE PRINT"Le joueur 2 gagne la partie"
5070 RETURN
6000 REM -----
6010 REM Jet de de
6020 REM -----
6030 REM
6040 LOCATE 3,21 : PRINT c1$+c2$+c2$+c2$+c3$
```

```

6050 LOCATE 3,22 : PRINT c4$+" "+c4$
6060 LOCATE 3,23 : PRINT c4$+" "+c4$
6070 LOCATE 3,24: PRINT c4$+" "+c4$
6080 LOCATE 3,25 : PRINT c5$+c2$+c2$+c2$+c6$
6090 de=INT(RND(1)*6)+1
6100 IF de=1 THEN LOCATE 5,23:PRINT c9$
6110 IF de=2 THEN LOCATE 4,22:PRINT c9$:LOCATE 6,24:PRINT c9$
6120 IF de=3 THEN LOCATE 4,22:PRINT c9$:LOCATE 5,23:PRINT c9$:LOCATE 6,24:PRINT c9$
6130 IF de=4 THEN LOCATE 4,22:PRINT c9$+" "+c9$:LOCATE 4,24:PRINT c9$+" "+c9$
6140 IF de=5 THEN LOCATE 4,22:PRINT c9$+" "+c9$:LOCATE 4,24:PRINT c9$+" "+c9$:LOCATE 5,23:PRINT c9$
6150 IF de=6 THEN LOCATE 4,22:PRINT STRING$(3,c9$):LOCATE 4,24:PRINT STRING$(3,c9$)
6160 RETURN
7000 REM -----
7010 REM Deplacement ou sortie d'un cheval
7020 REM -----
7030 REM
7040 IF so>0 THEN 7380 'Sortie d'un cheval
7050 IF ((jo=1) AND (p1=48)) OR ((jo=2) AND (p5=24)) THEN 7490
7060 IF ((jo=1) AND (c2=1)) OR ((jo=2) AND (d2=1)) THEN 7350 'Impossible de jouer
7070 FOR i=1 TO de
7080   IF jo=2 THEN 7230
7090   '
7100   ' - - - - -
7110   ' Deplacement du joueur 1
7120   ' - - - - -
7130   '
7140   IF i=1 THEN t(p1)=0
7150   LOCATE x(p1),y(p1):PEN 1
7160   IF t(p1)=2 THEN PEN 3:PRINT"1":PEN 1 ELSE PRINT"."
7170   p1=p1+1:IF p1=49 THEN p1=1
7180   LOCATE x(p1),y(p1):PEN 2:PRINT "1":PEN 1
7190   IF (i=de) AND (t(p1)<>0) THEN d2=1:p5=0 'Annulation joueur 2
7200   IF (i=de) THEN t(p1)=1
7210   GOTO 7340
7220   '
7230   ' - - - - -
7240   ' Deplacement du joueur 2
7250   ' - - - - -
7260   '
7270   IF i=1 THEN t(p5)=0

```

```

7280 LOCATE x(p5),y(p5):PEN 1
7290 IF t(p5)=1 THEN PEN 2:PRINT"1":PEN 1 ELSE PRINT"."
7300 p5=p5+1:IF p5=49 THEN p5=1
7310 LOCATE x(p5),y(p5):PEN 3:PRINT"1":PEN 1
7320 IF (i=de) AND (t(p5)<>0) THEN c2=1:p1=0 'Annulation j
oueur 1
7330 IF (i=de) THEN t(p5)=2
7340 NEXT i
7350 IF jo=1 THEN jo=2 ELSE jo=1
7360 RETURN
7370 '
7380 ' - - - - -
7390 ' Sortie d'un cheval
7400 ' - - - - -
7410 '
7420 so=0
7430 IF (jo=1) AND (t(1)=0) AND (c2<>0) THEN c2=c2-1:p1=1:t(
1)=1:LOCATE x(p1),y(p1)
7440 IF (jo=2) AND (t(25)=0) AND (d2<>0) THEN d2=d2-1:p5=25:
t(25)=2:LOCATE x(p5),y(p5)
7450 IF jo=1 THEN PEN 2 ELSE PEN 3
7460 PRINT"1"
7470 PEN 1
7480 GOTO 7350
7490 '
7500 ' - - - - -
7510 ' Etape finale pour un cheval
7520 ' - - - - -
7530 '
7540 IF (jo=1) AND (14-f1-de>=8) THEN LOCATE 19,14-f1:PEN 1:
PRINT".":LOCATE 19,14-f1-de:PEN 2:PRINT"1":f1=f1+de
7550 IF (jo=2) AND (f2+de+2<=8) THEN LOCATE 19,2+f2:PEN 1:PR
INT".":LOCATE 19,2+f2+de:PEN 3:PRINT"1":f2=f2+de
7560 IF (14-f1=8) OR (2+f2=8) THEN fin=1
7570 GOTO 7350

```

Lancez le programme. L'aire de jeu apparaît sur l'écran. Les deux écuries sont affichées en bleu et en rouge. Chacune contient quatre chevaux. Dans la version présentée, un seul cheval peut être déplacé sur l'aire de jeu.

Si vous le désirez (à titre d'exercice de programmation), modifiez le programme pour qu'il puisse gérer quatre chevaux dans chaque camp.

Une ligne de texte commente chaque partie du jeu. Sa couleur est celle de l'écurie courante (bleu ou rouge).

Au départ, la ligne doit contenir le texte suivant :

**Joueur 1, appuyez sur une touche**

Après l'appui sur une touche quelconque du clavier, un dé est jeté aléatoirement. Sa valeur apparaît en bas à droite de l'écran. Pour que le cheval puisse sortir de l'écurie, le dé doit avoir la valeur 6. Le joueur peut utiliser la valeur du dé ou passer son tour. Dans le premier cas, il doit appuyer sur **J**, et dans le second sur **P**. Comme nous le verrons, il est judicieux de passer son tour lorsque le cheval se trouve à proximité de la ligne finale et risque de la dépasser en utilisant le jet de dé.

Le jeu se poursuit en alternant entre le joueur 1 et le joueur 2. Lorsqu'un 6 est tiré, le cheval correspondant apparaît sur la case de départ **D**. Les futurs jets de dé le feront avancer vers la ligne finale (qui se trouve toujours une case avant la case de départ).

Le but du jeu est d'atteindre la case **A** avant l'adversaire. Pour y parvenir, il faut :


- arrêter le cheval devant la ligne finale. Vous comprenez maintenant pourquoi il peut être intéressant de passer son tour ;
- avancer de six cases en un ou plusieurs jets de dé. Lorsqu'un cheval est parvenu sur la case **A**, le programme prend fin et félicite le vainqueur.

*Remarque :*

Lorsqu'un cheval s'arrête sur une case où se trouvait un cheval, il renvoie ce dernier à l'écurie.

**LE PROGRAMME EN DÉTAIL**

La logique du programme apparaît dans l'ordinogramme ci-dessous :



Les premières lignes du programme activent les sous-programmes principaux :

**1040 GOSUB 2000** → Affichage du jeu

**1050 GOSUB 4000** → Jeu

**1060 GOSUB 5000** → Commentaires

Le sous-programme situé en 2000 affiche le jeu au début de chaque partie. Il utilise de nombreux caractères graphiques pour afficher le parcours et le dé. Ces caractères sont stockés dans les variables chaînes **c1\$** à **c9\$** lignes 2100 à 2160 pour faciliter leur manipulation :

**2100 c1\$=CHR\$(150)** 'Coin supérieur gauche

...

**2160 c9\$=CHR\$(227)** 'Point sur le dé

Plusieurs tableaux sont utilisés par le programme :

**T** contient la disposition des chevaux sur les quarante-huit cases de la périphérie du parcours ;

**TP** contient la position des chevaux sur les douze cases des deux lignes finales ;

**X** et **Y** contiennent la position des chevaux sur le parcours périphérique.

Ces tableaux sont définis lignes 2170 et 2180.

Le parcours est tracé à l'aide de nombreuses instructions **LOCATE** et **PRINT** lignes 2240 à 2510.

Le sous-programme se poursuit par le tracé des écuries lignes 2520 à 2880.

Les lignes 2890 à 3080 affichent le « tableau de bord ». Dans la version actuelle du jeu, seule la rubrique « Jet de dé » est exploitée.

Les tableaux **X** et **Y** sont ensuite initialisés avec des valeurs situées en **DATA** lignes 3140 à 3210. Ces deux tableaux permettent de connaître instantanément la position en ligne et colonne d'une case du parcours en fonction de son numéro.

Par exemple, les coordonnées de la case 23 se trouvent dans les variables **X(23)** et **Y(23)**.

Le sous-programme principal se trouve entre les lignes 4000 et 4310. Il est responsable du déroulement du jeu.

Les lignes 4070 à 4090 effacent l'ancien jet de dé.

Les lignes 4110 et 4120 affichent un commentaire en fonction du joueur courant.

La ligne 4140 attend que le joueur courant appuie sur une touche du clavier :

```
4140 a$=INKEY$:IF a$="" THEN 4140
```

La ligne 4170 active le sous-programme de jet de dé situé en 6000 :

```
4170 GOSUB 6000
```

Les lignes 4180 et 4190 font sortir un cheval lorsque les conditions sont réunies :

- jet de dé égal à 6 ;
- le cheval n'est pas déjà dehors.

Les lignes suivantes demandent au joueur courant d'utiliser le jet de dé ou de passer son tour :

```
4220 PRINT "Tapez J pour jouer et P pour passer"
```

```
...
```

```
4290 IF a$="J" THEN GOSUB 7000 'Déplacement
```

Si aucun des joueurs n'est arrivé sur la case **A**, la ligne 4300 renvoie à la boucle de jeu :

```
4300 IF fin=0 THEN 4100
```

Le sous-programme situé entre les lignes 5000 et 5070 est activé lorsqu'un des joueurs a gagné. Il se contente d'afficher le nom du vainqueur.

Le sous-programme qui affiche chaque jet de dé se trouve entre les lignes 6000 et 6160.

Les contours du dé sont tracés lignes 6040 à 6080 à l'aide de caractères semi-graphiques stockés dans les variables **c1\$** à **c6\$**.

Le caractère utilisé pour matérialiser chaque point du dé se trouve dans **c9\$**. Le tirage aléatoire se fait ligne 6090. Le résultat est un nombre entier compris entre 1 et 6 :

```
6090 de=INT(RND(1)*6)+1
```

En fonction du nombre tiré, un ou plusieurs points sont affichés sur le dé lignes 6100 à 6150.

Le sous-programme situé entre les lignes 7000 et 7570 est responsable de la sortie et du déplacement des chevaux. Lorsque les conditions sont réunies pour la sortie d'un cheval, la variable **so** est non nulle. La ligne 7040 effectue alors un débranchement en 7380 :

```
7040 IF so>0 THEN 7380 'Sortie d'un cheval
```

Dans le cas contraire, le programme teste si le cheval courant se trouve sur la ligne finale. Si oui, il effectue un débranchement en 7490 :

```
7050 IF ((jo=1) AND (p1=48)) OR ((jo=2) AND (p5=24)) THEN 7490
```

La ligne 7060 teste si le cheval se trouve dans l'écurie. Si oui, il est impossible de le déplacer :

```
7060 IF ((jo=1) AND (c2=1)) OR ((jo=2) AND (d2=1)) THEN 7350
```

Lorsque tous ces tests ont échoué, le cheval peut être déplacé. Les lignes 7140 à 7210 déplacent le cheval du premier joueur, et les lignes 7270 à 7330 le cheval du second joueur. La ligne 7350 fait passer la main :

```
7350 IF jo=1 THEN jo=2 ELSE jo=1
```

Comme nous l'avons dit plus haut, la sortie d'un cheval se fait ligne 7420. Les actions effectuées sont les suivantes :

- diminution du nombre de chevaux dans l'écurie (c2-1 pour le joueur 1, d2-1 pour le joueur 2) ;
- positionnement du cheval sur la case 1 pour le premier joueur et sur la case 25 pour le second joueur ;
- affichage du cheval sur la case départ.

Les dernières lignes du programme gèrent les déplacements sur la ligne finale. Tout jet de dé faisant dépasser la ligne d'arrivée ( $14-f1-de \geq 8$  ou  $f2+de+2 \leq 8$ ) est refusé.

La ligne 7560 initialise la variable fin à 1 lorsqu'un cheval a atteint la case A :

```
7560 IF (14-f1=8) OR (2+f2=8) THEN fin=1
```

## Variante du jeu de chevaux :

### Le jeu de chevaux à quatre joueurs

#### COMMENT UTILISER LE PROGRAMME

L'utilisation du programme est la même que celle décrite précédemment, si ce n'est que maintenant, le nombre de chevaux dans chaque camp est égal à quatre.

#### LES DIFFÉRENCES ENTRE LES DEUX VERSIONS DU PROGRAMME

Les premières lignes du programme (lignes 1000 à 3220) n'ont pas changé. Seules diffèrent les instructions responsables de la gestion d'une partie (ligne 4000 à 5280).

Etudions les différences par rapport à la version « monocheval » :

- le nombre de chevaux (variables **c2** et **d2**) passe de 1 à 4 (ligne 4060) ;
- les chevaux à déplacer sont repérés par leur numéro ;
- le test de sortie d'un cheval (lignes 4180, 4190 et 4310) utilise la variable **so** d'une manière différente ;
- le déplacement des chevaux doit maintenant tenir compte du numéro du cheval à déplacer (lignes 4600 à 4960) ;
- la sortie d'un cheval utilise les variables **p**, **c2** et **d2** d'une manière mieux adaptée à la gestion de plusieurs chevaux.

```

1000 REM =====
1010 REM Jeu de chevaux a deux joueurs
1020 REM =====
1030 REM
1040 GOSUB 2000 'Affichage du jeu
1050 GOSUB 4000 'Jeu
1060 GOSUB 4350 'Commentaires sur la partie
1070 END
1080 '
2000 REM -----
2010 REM Affichage du jeu de depart
2020 REM -----
2030 REM
2040 MODE 1
2050 '
2060 ' - - - - -
2070 ' Caracteres graphiques utilises
2080 ' - - - - -
2090 '
2100 c1$=CHR$(150) 'Coin superieur gauche
2110 c2$=CHR$(154) 'Trait horizontal
2120 c3$=CHR$(156) 'Coin superieur droit
2130 c4$=CHR$(149) 'Trait vertical
2140 c5$=CHR$(147) 'Coin inf2rieur gauche
2150 c6$=CHR$(153) 'Coin inferieur droit
2160 c9$=CHR$(227) 'point sur le de
2170 DIM t(48),tp(12) 'Tableaux de jeu
2180 DIM x(48),y(48) 'Tableaux des positions
2190 '
2200 ' - - - - -
2210 ' Trace du circuit
2220 ' - - - - -
2230 '
2240 PEN 1
2250 s1$=c1$+STRING$(7,c2$)+c3$
2260 s2$=c4$+" . . . "+c4$
2270 s3$=c1$+STRING$(9,c2$)+c6$+" . . . "+c5$+STRING$(9,c2$)
+c3$
2280 s4$=c5$+STRING$(9,c2$)+c3$+" . . . "+c1$+STRING$(9,c2$)
+c6$
2290 s5$=c4$+" . . . "+c4$
2300 s6$=c5$+STRING$(7,c2$)+c6$
2310 s7$=c4$+" . . . . . "+c4$
2320 LOCATE 15,1
2330 PRINT s1$
2340 FOR i=1 TO 4
2350   LOCATE 15,i+1
2360   PRINT s2$
2370 NEXT i
2380 LOCATE 5,6

```

```
2390 PRINT s3$
2400 FOR i=1 TO 3
2410 LOCATE 5,i+6
2420 PRINT s7$
2430 NEXT i
2440 LOCATE 5,10
2450 PRINT s4$
2460 FOR i=1 TO 4
2470 LOCATE 15,i+10
2480 PRINT s5$
2490 NEXT i
2500 LOCATE 15,15
2510 PRINT s6$
2520 '
2530 ' - - - - -
2540 ' Trace des ecuries
2550 ' - - - - -
2560 '
2570 PEN 2
2580 s1$=c1$+STRING$(4,c2$)+c3$
2590 s2$=c4$+SPACE$(4)+c4$
2600 s3$=c5$+STRING$(4,c2$)+c6$
2610 LOCATE 7,11
2620 PRINT s1$
2630 FOR i=1 TO 3
2640 LOCATE 7,i+11
2650 PRINT s2$
2660 NEXT i
2670 LOCATE 7,15
2680 PRINT s3$
2690 LOCATE 8,12
2700 PRINT"o o"
2710 LOCATE 8,14
2720 PRINT"o o"
2730 PEN 3
2740 LOCATE 26,1
2750 PRINT s1$
2760 FOR i=1 TO 3
2770 LOCATE 26,i+1
2780 PRINT s2$
2790 NEXT i
2800 LOCATE 26,5
2810 PRINT s3$
2820 LOCATE 27,2
2830 PRINT"x x"
2840 LOCATE 27,4
2850 PRINT"x x"
2860 PEN 1 : LOCATE 19,8 : PRINT"A"
2870 PEN 2 : LOCATE 17,14 : PRINT"D"
2880 PEN 3 : LOCATE 21,2 : PRINT"D"
```

```

2890 '
2900 ' - - - - -
2910 ' Tableau de bord
2920 ' - - - - -
2930 '
2940 LOCATE 1,18
2950 PEN 1 : PRINT STRING$(40,c2$)
2960 LOCATE 1,20
2970 PRINT"Jet de des"
2980 LOCATE 17,20
2990 PEN 2 : PRINT"Joueur 1"
3000 LOCATE 31,20
3010 PEN 3 : PRINT"Joueur 2"
3020 PEN 1
3030 FOR i=1 TO 2
3040     FOR j=1 TO 4
3050         LOCATE 13+(i-1)*15,j+19
3060         PRINT c4$
3070     NEXT j
3080 NEXT i
3090 '
3100 ' - - - - -
3110 ' Memorisation des cases de deplacement
3120 ' - - - - -
3130 '
3140 FOR i=1 TO 48
3150     READ x(i),y(i)
3160 NEXT i
3170 DATA 17,14,17,13,17,12,17,11,17,10,17,9,15,9,13,9,11,9,
9,9,7,9
3180 DATA 7,8,7,7,9,7,11,7,13,7,15,7,17,7,17,6,17,5,17,4,17,
3,17,2
3190 DATA 19,2,21,2,21,3,21,4,21,5,21,6,21,7,23,7,25,7,27,7,
29,7,31,7
3200 DATA 31,8,31,9,29,9,27,9,25,9,23,9,21,9,21,10,21,11,21,
12,21,13
3210 DATA 21,14,19,14
3220 RETURN
4000 REM -----
4010 REM Deroulement d'une partie
4020 REM -----
4030 REM
4040 PEN 1
4050 jo=1
4060 c2=4 : d2=4 'Quatre chevaux dans chaque camp
4070 FOR i=21 TO 25
4080     LOCATE 3,i:PRINT SPACE$(10);
4090 NEXT i
4100 LOCATE 5,17
4110 IF jo=1 THEN PEN 2:PRINT"Joueur 1, appuyez sur une touc

```

```

he"
4120 IF jo=2 THEN PEN 3:PRINT"Joueur 2, appuyez sur une touc
he"
4130 PEN 1
4140 a$=INKEY$:IF a$="" THEN 4140
4150 LOCATE 5,17
4160 PRINT SPACE$(32)
4170 GOSUB 4430 'Jet de de
4180 IF (de=6) AND (c2<>0) AND (jo=1) THEN so=5-c2 'Sortie c
heval bleu
4190 IF (de=6) AND (d2<>0) AND (jo=2) THEN so=5-d2 'Sortie c
heval rouge
4200 LOCATE 5,17
4210 PEN jo+1
4220 PRINT"Cheval a deplacer ou P pour passer"
4230 PEN 1
4240 a$=INKEY$:IF a$="" THEN 4240
4250 a$=UPPER$(a$)
4260 so$=a$
4270 a=VAL(a$)
4280 IF so<>a THEN so=0
4290 LOCATE 1,17
4300 PRINT SPACE$(40)
4310 IF a$="P" THEN IF jo=1 THEN jo=2 ELSE jo=1
4320 IF ((a<=5-c2) AND (a<>0) AND (jo=1)) OR ((a<=5-d2) AND
(a<>0) AND (jo=2)) THEN GOSUB 4600
4330 IF fin=0 THEN 4100 'Boucle de jeu
4340 RETURN
4350 REM -----
4360 REM Commentaire sur la partie
4370 REM -----
4380 REM
4390 CLS
4400 PEN 1
4410 IF (14-f1=8) THEN PRINT" Le joueur 1 gagne la partie" E
LSE PRINT"Le joueur 2 gagne la partie"
4420 RETURN
4430 REM -----
4440 REM Jet de de
4450 REM -----
4460 REM
4470 LOCATE 3,21 : PRINT c1$+c2$+c2$+c2$+c3$
4480 LOCATE 3,22 : PRINT c4$+" "+c4$
4490 LOCATE 3,23 : PRINT c4$+" "+c4$
4500 LOCATE 3,24: PRINT c4$+" "+c4$
4510 LOCATE 3,25 : PRINT c5$+c2$+c2$+c2$+c6$
4520 de=INT(RND(1)*6)+1
4530 IF de=1 THEN LOCATE 5,23:PRINT c9$
4540 IF de=2 THEN LOCATE 4,22:PRINT c9$:LOCATE 6,24:PRINT c9
$

```

```

4550 IF de=3 THEN LOCATE 4,22:PRINT c9$:LOCATE 5,23:PRINT c9
$:LOCATE 6,24:PRINT c9$
4560 IF de=4 THEN LOCATE 4,22:PRINT c9$+" "+c9$:LOCATE 4,24:
PRINT c9$+" "+c9$
4570 IF de=5 THEN LOCATE 4,22:PRINT c9$+" "+c9$:LOCATE 4,24:
PRINT c9$+" "+c9$:LOCATE 5,23:PRINT c9$
4580 IF de=6 THEN LOCATE 4,22:PRINT STRING$(3,c9$):LOCATE 4,
24:PRINT STRING$(3,c9$)
4590 RETURN
4600 REM -----
4610 REM Deplacement ou sortie d'un cheval
4620 REM -----
4630 REM
4640 IF so>0 THEN 4980 'Sortie d'un cheval
4650 IF ((jo=1) AND (p(a)=48)) OR ((jo=2) AND (p(a+4)=24)) T
HEN 5200 'Etape finale
4660 IF ((jo=1) AND (p(a)=0)) OR ((jo=2) AND (p(a+4)=0)) THE
N 4950 'Impossible de jouer
4670 FOR i=1 TO de
4680   IF jo=2 THEN 4830
4690   '
4700   ' - - - - -
4710   ' Deplacement du joueur bleu
4720   ' - - - - -
4730   '
4740   IF i=1 THEN t(p(a))=0
4750   LOCATE x(p(a)),y(p(a)):PEN 1
4760   IF t(p(a))=0 THEN PRINT"."
4770   p(a)=p(a)+1:IF p(a)=49 THEN p(a)=1
4780   IF (t(p(a))=0) OR (i=de) THEN LOCATE x(p(a)),y(p(a)):
PEN 2:PRINT so$:PEN 1
4790   IF (i=de) AND (t(p(a))<>0) THEN p(t(p(a)))=0 'Annulat
ion joueur 2
4800   IF (i=de) THEN t(p(a))=a
4810   GOTO 4940
4820   '
4830   ' - - - - -
4840   ' Deplacement du joueur 2
4850   ' - - - - -
4860   '
4870   IF i=1 THEN t(p(a+4))=0
4880   LOCATE x(p(a+4)),y(p(a+4)):PEN 1
4890   IF t(p(a+4))=0 THEN PRINT"."
4900   p(a+4)=p(a+4)+1:IF p(a+4)=49 THEN p(a+4)=1
4910   IF (t(p(a+4))=0) OR (i=de) THEN LOCATE x(p(a+4)),y(p(
a+4)):PEN 3:PRINT so$:PEN 1
4920   IF (i=de) AND (t(p(a+4))<>0) THEN p(t(p(a+4)))=0 'Ann
ulation joueur 1
4930   IF (i=de) THEN t(p(a+4))=a+4
4940 NEXT i

```

```

4950 IF jo=1 THEN jo=2 ELSE jo=1
4960 RETURN
4970 '
4980 ' - - - - -
4990 ' Sortie d'un cheval
5000 ' - - - - -
5010 '
5020 IF jo=2 THEN 5080
5030 IF so=1 THEN LOCATE 8,12
5040 IF so=2 THEN LOCATE 11,12
5050 IF so=3 THEN LOCATE 8,14
5060 IF so=4 THEN LOCATE 11,14
5070 GOTO 5120
5080 IF so=1 THEN LOCATE 27,2
5090 IF so=2 THEN LOCATE 30,2
5100 IF so=3 THEN LOCATE 27,4
5110 IF so=4 THEN LOCATE 30,4
5120 PRINT" "
5130 IF (jo=1) AND (t(1)=0) AND (c2<>0) THEN c2=c2-1:p(a)=1:
t(1)=so:LOCATE x(p(a)),y(p(a))
5140 IF (jo=2) AND (t(25)=0) AND (d2<>0) THEN d2=d2-1:p(a+4)
=25:t(25)=so+4:LOCATE x(p(a+4)),y(p(a+4))
5150 IF jo=1 THEN PEN 2 ELSE PEN 3
5160 PRINT RIGHT$(STR$(so),1)
5170 PEN 1
5180 so=0
5190 GOTO 4950
5200 '
5210 ' - - - - -
5220 ' Etape finale pour un cheval
5230 ' - - - - -
5240 '
5250 IF (jo=1) AND (14-f1-de>=8) THEN LOCATE 19,14-f1:PEN 1:
PRINT".":LOCATE 19,14-f1-de:PEN 2:PRINT"1":f1=f1+de
5260 IF (jo=2) AND (f2+de+2<=8) THEN LOCATE 19,2+f2:PEN 1:PR
INT".":LOCATE 19,2+f2+de:PEN 3:PRINT"1":f2=f2+de
5270 IF (14-f1=8) OR (2+f2=8) THEN fin=1
5280 GOTO 4950

```

